
Part 10: Recycling Code 
Good developers know that creating reusable code is an important part of application 

development, and always seek practical ways for doing so while developing. Reusable code 

 Saves valuable development time, allowing you avoid "rebuilding the wheel" for every 
app or page 

 Reduces testing time as once the code is tested once, it reduces the changes of it 
breaking when testing other uses of it 

 Makes it easier to share code between applications and/or developers 

 Makes it easier to implement global changes: once you change to one piece of code, it 
immediately and instantly propagates across the entire app 

ColdFusion offers several options to help you with code reuse: 

Method Usage 

Include pages using <cfinclude> Primarily for static content that has to be included in 
multiple pages, but is not changed frequently. Shared 
scope with calling page. 

<cfinclude template="inctable.cfm"> 

Custom tags  Similar to <cfinclude> but includes content that can be 
modified with tag attributes, and more dynamic content. 
Protected scope. 

<cf_header title= "Page Title"> 

User-Defined Functions (UDFs)  Use when you want to pass a value to a function and 
get a value back. 

<cfinclude template="CVFunctions.cfm"> 
<cfoutput>Today’s Date: #date()# </cfoutput> 

ColdFusion Components (CFCs)  Use to encapsulate business logic that you want to 
reuse across pages or other applications, such as Flash, 
queries, etc. 

Custom Tags 

Developers can extend the CFML tags available to them by creating custom tags. These tags 

are called in much the same way as any other CFML tag. There are three varieties of custom 

tags: 

 Custom tags that contain CFML code 

 Java custom tags (CFX) 

 C++ DLLs that contain functions you wish to call (CFX) 

As CFX requires one to know Java and/or C++, we will be looking purely at regular custom 

tags containing regular CFML code. CFML custom tags can be called from any page or 

template. Unlike using a <cfinclude> tag to reuse code, custom tags have a "protected" scope, 

so that variables created in the custom tag does not conflict with variables in the calling page, 

and visa versa. However, you can explicitly pass values to and from the tags. This makes it 

much easier to create more modular systems for proper code reuse and portability. 

A custom tag is written and saved just like any other ColdFusion file with a ".cfm" extension. To 

call it from another page, you would use the prefix "cf_" followed by the page name. For 

instance, to use a file named foomanchu.cfm, use <cf_foomanchu>. Like any other CFML tag, 

you enclose custom tags within brackets (< >).   

Tip: If a custom tag is a C++ or Java type, you would use the prefix "cfx_" instead. 

Remember when ColdFusion encounters a <cfinclude> it basically performs a virtual 

copy/paste of the code from that template where the include is called, and continues 

processing both pages as one? With a custom tag call, ColdFusion stops processing the page 

it is on, finds the related file, process the code there, and then returns to its calling page to 

continue processing. 

Figure 1: Processing a custom tag call 

You may notice that when you call a custom tag, you don't give ColdFusion any kind of path 

information like you do with a <cfinclude>. Instead, ColdFusion searches for the tag in 

specifically places in a specific order: 

1. The same directory of the calling page 

2. The cfusion\CustomTags directory and any subdirectories under it 

3. Any other custom tag paths specifically set in the Extensions > Custom Tags menu in 
the ColdFusion Administrator 

If it cannot find the file in any of these places, it will throw an error. While ColdFusion itself is 

case insensitive, when working on UNIX/Linux systems, always name your custom tags with 

lowercase names.  

Scoping with Custom Tags 

Just like any other tags, you can use attributes with your custom tags, which can then be 

accessed by the tag when it is processing. To assign attributes to a custom tag, you simply 

append attribute=value pairs to it inside the body of the custom tag. 

<cf_foo attribute1="value1" attribute2="value2"...> 

Of course, just passing along attributes is fairly useless if you do not also build your tag to 

accept and process them. Attributes passed to a custom tag are stored in the Attributes scope. 

So if we had a custom tag call of:  

<cf_mytag Firstname="Thadeus" Lastname="Jones"> 

Within the mytag.cfm file, we would refer to the passed values using Attributes.Firstname and 

Attributes.Lastname. The custom tag page can also access other variables that were set in the 

calling page using the Caller scope, however this is a bad way to pass information to a tag as it 

reduces portability and makes the tag dependant on its caller page. Using attributes keeps 

your tags flexible. Custom tags can, of course, access variables in other scopes, such as URL, 

Form, CGI, Session, and Application, but again, keep flexibility in mind before relying on any of 

these scopes. 

Variables created within the custom tag page are inaccessible to other pages, and they are 

deleted once the tag has finished running. So, if you need to pass information from a custom 

tag back to the caller, you must set those attributes using the Caller scope. For example, if we 

had a custom tag called getmd that needed to pass back a variable called Doctor, we would 

use: 

<cfset Caller.Doctor="Mr. Doctor " & Attributes.Name> 

If the variable Doctor does not exist in the calling page, it is created. If it does exist, it is 

overwritten, so care should be used with variable names. One common technique for ensuring 

ultimate flexibility in a custom tag is to have it accepted an attribute that contains the name of 

the variable it should use for passing back results. For example, the calling page passes 

returnHere as the name of the variable to use to pass back results:  

<cf_mytag resultName="returnHere"> 

Then, in mytag.cfm, the custom tag passes back its results using the following code:  

<cfset "Caller.#Attributes.resultName#" = result>  

It is important to document your custom tags to indicate what attributes are acceptable, 

including which ones are required or not. A well-formed custom tag will test that its required 

and optional attributes exist and use them appropriately, and return appropriate error 

messages for required attributes that are missing using appropriate conditional logic. 

Tip: Custom tags can work recursively, as in a custom tag can call itself without breaking itself. 

This can be useful in dealing with things like recursive categories and breadcrumbs.  

End tags 

Custom tags can have ending tags, or be self-closing, however unlike most ColdFusion tags, 

you do not want to self-close custom tags by default. When ColdFusion encounters a closing 

tag or self-closing custom tag, it will call the tag twice, one for the start and once for the end 

tag (or use a self-closing tag), so you need to make sure you use conditional logic to account 

for this. You can refer to a special variable within the custom tag, thisTag.ExecutionMode, 

which will return Start, End, or Inactive values depending on its state. Let's say we had a 

custom tag called <cf_bold> that contained: 

<cfif thisTag.ExecutionMode is 'start'> 

<!--- Start tag processing ---> 

<B> 

<cfelse> 

<!--- End tag processing ---> 

</B> 

</cfif> 

If we called it using: 

<cf_bold>This is bold text</cf_bold> 

The result would be: 

<b>This is bold text</b> 

You can also put content between the opening/closing tags.  

<cf_happybirthdayMessge name="Ellen Smith" birthDate="June 8, 1993"> 

    <p> Happy Birthday Ellen!</p> 

    <p> May you have many more!</p> 

</cf_happybirthdayMessge> 

You can access the body text placed between the opening and ending tags using the 

thisTag.GeneratedContent variable. This variable contains all body text passed to the tag, 

making it available for you to modify if desired. The contents of the thisTag.GeneratedContent 

variable are returned to the browser as part of the tag's output. However, the variable is always 

empty during the processing of a start tag. Any output generated during start tag processing is 

not considered part of the tag's generated content. The variable is only populated when during 

end tag processing. 

Alternate calling methods 

In addition the method described above, custom tags can be called using the <cfmodule> tag 

or with the <cfimport> tag. The <cfmodule> tag allows you specify the location of your custom 

tag (useful when you have multiple instances of a tag on a server) and to use a variable to 

dynamically declare the custom tag to run. 

<cfmodule  

attributeCollection = "collection structure" 

attribute_name1 = "valuea" 

attribute_name2 = "valueb" 

name = "tag name" 

template = "path" 

...> 

When using <cfmodule> you can pass attributes in name=value pairs, the same as with a 

regular call, or in a structure of name=value through the attributeCollection attribute. You can 

also use both within the same custom tag calling. Within the custom tag itself, the attributes 

passed by attributeCollection are saved in individual name=value pairs, without any indication 

of their being grouped by the caller tag. 

The name and template attributes of the <cfmodule> tag are mutually exclusive, you can only 

use one or the other. With the name attribute, the tag must be under the ColdFusion root 

directory and is referred to using dot (.) notation. This example points to a custom tag called 

GetUserOptions under the Forums40 directory under the ColdFusion root directory:  

<cfmodule name = ".Forums40.GetUserOptions">   

If you use the template attribute, the path can be relative, or absolute from a ColdFusion 

mapping. Physical paths are not allowed. 

With the <cfimport> tag, you can import custom tags from a directory as a tag library. When 

using this tag, it will import all of the tags from the specified directory: 

<cfimport prefix="mytags" taglib="myCustomTags"> 

You then call the custom tags using the prefix set in the tag while importing. You can also still 

include name=value attributes when doing this call. 

<mytags:customTagName> 

<mytags:custom_tag_name attribute1=val_1 attribute2=val_2> 

An advantage of using the <cfimport> tag is that it facilitates defining a directory structure for 

your custom tags, allowing you to organize them by category or some other method. For 

example, you can put all security tags in one directory, and all interface tags in another. You 

then import the tags from each directory and give them a different prefix: 

<cfimport prefix="security" taglib="securityTags"> 

<cfimport prefix="ui" taglib="uiTags"> 

<security:validateUser name="Bob"> 

<ui:greeting name="Bob"> 

This can help make reading your code easier, as the prefixes make it easier to identify both the 

location and basic function of your custom tags. 

User-Defined Functions 

User-defined functions, or UDFs, allow you to create your own reusable functions that can 

accept arguments and return values. They differ from custom tags in their method of calling 

and general purpose. You call a UDF in the same fashion as a regular ColdFusion function: 

function_name(argument1, argument2,..) 

You can call them anywhere you can use a normal expression or function, including: 

 Within a variable using pound signs (#) and within a <cfoutput> tag or the <cfquery> 
tag, within a cfset without pound signs 

 In a ColdFusion script (learn more about the <cfscript> tag in the Advanced ColdFusion 
Development course) 

 As a value for an attribute in a tag 

 As an argument to another function 

Just as with ColdFusion's built-in functions, a UDF's argument names are not specified in the 

function call, making the order in which you list the arguments important. In this example, the 

code called a UDF called amortization, passing it appropriate arguments and displaying the 

return value. 

<cfoutput> 

#amortization("140000", "7", "360")# 

</cfoutput> 

When you call a UDF, ColdFusion will look for it in two locations: 

 Within the page that is calling it 

 Within a page that is included using <cfinclude> 

UDFs can be created using CFScript or using the <cffunction> tag. The latter type forms the 

basis of ColdFusion components, allowing them to be called in more advanced ways. UDFs 

generally require less overhead than a custom tag, but their use can be limited. Typically, they 

are used to expand upon existing functions to perform things like: 

 Data manipulation routines, such as a function to reverse an array  

 String and date and time routines, such as a function to determine whether a string is a 
valid IP address  

 Mathematical calculation routines, including standard trigonometric and statistical 
operations or calculating loan amortization  

 Routines that call functions externally, for example using COM or CORBA, such as 
routines to determine the space available on a Windows file system drive  

Introducing ColdFusion Components (CFCs) 

ColdFusion components (CFCs) are groups of functions and properties related to a particular 

category or purpose. For instance, in a shopping cart CFC, you might have functions for 

addItem, calculateTotal, computeTax, and a cartItems property. All CFC functions, or methods, 

and properties are placed within a file with a .cfc extension, such as ShoppingCart.cfc. The 

basic syntax of a CFC is: 

<cfcomponent> 

<cffunction name="myFunction" access="public" 

returntype="string"> 

<cfargument name="myArgument" type="string" required="yes" 

/> 

<cfset myResult="foo"> 

<cfreturn myResult> 

</cffunction> 

</cfcomponent> 

A pair of <cfcomponent> tags surrounds the CFC's code, within which exist one or more 

functions. The <cffunction> tag defines a function, declaring its name, how it can be accessed, 

and what type of value it will return, if any. If the function accepts arguments, these are then 

declared using the <cfargument> tag to define each argument, the acceptable data type for 

that argument, and if it is required or not. Values passed to functions are stored and referred to 

using the Arguments scope. 

Within the <cffunction> tag block, it is common to find <cfset> and <cfquery> tags as functions 

are often used to process variables and queries. If the function needs to return some type of 

value to the calling page, use the <cfreturn> tag. The value returned must be of the same type 

declared in the opening <cffunction> tag. 

When you wish to use a component, you must "invoke" through the <cfinvoke> tag, like so: 

<cfinvoke component="componentName" method="methodName" 

returnVariable="variableName"> 

<cfinvokeargument name="parameter1Name" value="anyValue"> 

</cfinvoke> 

The <cfinvokeargument> tag is used to pass variables to the CFC, corresponding to those 

defined within the <cffunction> tag. Again, the values passed with the <cfinvokeargument> 

tags must match the types declared within the function itself. Returning variables and inputting 

arguments are optional. A CFC method can also create its own output by using the <cfoutput> 

tag. 

Note: CFC instances may also be created using the <cfobject> tag or the creationObject() 

function but this is less commonly used. 

When invoking a component, you must tell ColdFusion where to find it, if the CFC is not in the 

same directory as the calling template. This is done using "package names" which are similar 

to directory paths, but declared using dot (.) notation. For example, to invoke a CFC located at 

C:\ColdFusion8\wwwroot\myLibrary\components\ShoppingCart.cfc, you would use: 

<cfinvoke component="myLibrary.components.ShoppingCart" method="add" 

returnVariable="myCart"> 

<cfinvokeargument name="productId" value="13"> 

<cfinvokeargument name="quantity" value="3"> 

<cfinvokeargument name="price" value="12.50"> 

</cfinvoke> 

The component attribute value, myLibrary.components.ShoppingCart, is the package name. 

Part 11: Securing Your Apps 
Like it or not, security is an important element a developer MUST keep in while developing and 

maintaining ColdFusion applications, and should always be one of their top concerns. Web 

applications are frequently used to collect personal information about user, including social 

security numbers, phone numbers, addresses, maiden names, passwords, and credit card 

information. It is your responsibility to ensure your applications are developed in such a way 

that the information is kept as secure as possible. One method of securing a web application is 

by using application and session variables. Application variables are used for sharing 

information across an application that is user independent, such as the application name, data 

source names, paths to directories, etc. They persist for all application users for a defined 

period of time or until the application is restarted. Session variables are more effective for 

securing information specific to a single browser (not necessarily user), with the information 

maintained until the user logs out or the browser is closed. 

Some commonly uses for session variables include: 

 Validate user authentication at login and maintain that authentication throughout the 
session. 

 Personalize the user’s experience with preferences like weather and stock feeds, and 
individual information like their name and user history. 

 Maintain information about the user’s session, like shopping cart data. 

When a user browsers the web, the HTTP protocol creates a new connection for each page 

request. Because each request is independent of all prior interactions, any variables and flags 

set during one page request are unavailable on the next request, unless they are persisted. 

When trying to authenticate users, you must take this stateless environment into account and: 

 Authenticate them on first access by displaying a login page to them. 

 Discontinue access to an application after a predetermined time of inactivity. 

 Secure each page to be sure they cannot bookmark a page and circumvent the login. 

The Application Framework 

To enable the application framework, one of two specialize files is required: 

 Application.cfc 

 Application.cfm 

Application.cfc is the preferred method of initializing the application framework, but 

Application.cfm might be found when dealing with pre-ColdFusion MX legacy code. 

Every ColdFusion page request involves the following steps: 

1. ColdFusion first looks for an Application.cfc file in the same directory as the requested 
page. If it finds the file, it will invoke it. 

2. If ColdFusion does not find an Application.cfc file, it looks for an Application.cfm file. If it 
finds this file, it will process it. 

3. ColdFusion will look up the entire file system tree of the website for any Application file 
and will run this file before processing the originally requested file. 

4. If no Application file is found, ColdFusion will process the originally requested file. 

All pages affected by a particularly Application file, or a set of related Application files, are 

considered to be a single "web application" by the ColdFusion server. As code within an 

Application file runs on every page request, it is the ideal location for commonly shared code, 

such as headers, footers, and, of course, login security tests. 

For security reasons, you cannot prevent the ColdFusion Server from invoking the Application 

framework file closest to any requested page. You can, however, override an Application file in 

a folder above the requested page by placing a different Application file in the same folder as 

the requested page. This can be useful in separating the user-side of a web application from a 

web-based administration side into two distinct applications. 

As noted, Application.cfm is now considered deprecated and should not be used in newer 

applications. Older applications should be transitioned to Application.cfc as soon as possible 

as part of its on-going maintenance. Any code placed within the Application.cfm file will run at 

the beginning of every request for a page within the same directory or any subdirectories under 

the directory it resides in, unless those directories have their own Application file.  

Unlike the Application.cfm file, the Application.cfc file allows developers to define code to 

execute at defined points during the operation of the application. These points are referred to 

as "events"; within the Application.cfc, one can define functions with special names to run 

when these events occur: “application event handlers”. 

Event  Occurs  

onApplicationStart  At first request following server re-start or application timeout  

onApplicationEnd  At server shutdown or application timeout  

onSessionStart  At first request of new user session  

onSessionEnd  At session timeout  

onRequestStart  At beginning of every page request  

onRequestEnd  At end of every page request  

onRequest  At beginning of every page request, and supports filtering of 
requested page content  

onError When an error occurs during a page request 

 

So, if we wanted to create an application start event handler, we would create a <cffunction> 

block named "onApplicationStart" within the Application.cfc file. 

<cfcomponent> 

<cffunction name="onApplicationStart"> 

... 

</cffunction> 

</cfcomponent> 

This event handler runs once during the first request of an application: following a server 

restart or after the application time-out has expired.  

The onRequestStart event handler runs at every page request, just before the requested page 

is processed. It is useful for user authorization (login handling), and for request-specific 

variable initialization, such as gathering performance statistics. If you use the onRequestStart 

method and do not use the onRequest method, ColdFusion will automatically process the 

request when it has finished processing the onRequestStart code. This event handler requires 

no arguments. 

<cffunction name="onRequestStart"> 

... 

</cffunction> 

In this example, we use onRequestStart and some CFScript to stop requests from 1-3 am for 

site maintenance: 

<cffunction name="onRequestStart">  

 <cfargument name="requestname" required="true" /> 

 <cfscript> 

  if ((Hour(now()) gt 1) and (Hour(now()) lt 3)) { 

   WriteOutput("The system is undergoing periodic 

maintenance. Please return after 3:00 AM Eastern time."); 

   return false; 

  } else { 

   this.start=now(); 

   return true; 

  } 

 </cfscript> 

</cffunction> 

The onRequest event handler runs on each page request, after the onRequestStart event 

handler and before the requested page is displayed. It takes one required argument, 

nameattribute, which contains the name of the requested template. 

<cfcomponent> 

<cffunction name="onRequest"> 

<cfargument name="targetTemplate" type="String" 

required="true" /> 

... 

</cffunction> 

</cfcomponent> 

You can use this name for filtering purposes, or for using conditional logic to determine which 

page is displayed. Unlike onRequestStart, the onRequest handler intercepts the request, and 

stops processing when done. Therefore, if you define an onRequest event handler, you must 

include the requested template, as shown here, or the requested page will not display. At 

minimum, you must have something to include the calling page, such as: 

<cfinclude template="#Arguments.targetTemplate#"> 

You can also use it for filtering purposes, by wrapping the <cfinclude> in a <cfsavecontent> tag 

then performing actions against the saved content. 

<cfsavecontent variable="content"> 

<cfinclude template=#Arguments.targetPage#> 

</cfsavecontent> 

<cfoutput> 

#replace(content, "report", "MyCompany Quarterly Report", "all")# 

</cfoutput> 

Another key difference between onRequestStart and onRequest is that the latter can access 

the Variables (local) scope of the calling page, however because it runs before the page, it 

can't read variables from it, only set them unless you use the filtering method above. 

The onRequestEnd event handler specifies code that should run at the end of the request 

before completing it, making it the ideal place for including common footers. If you use the 

onRequest event handler, then onRequestEnd will also have access to the Variables scope, 

otherwise it does not, though you can use the Request scope to share data with the requested 

page. 

As you might guess, OnSessionStart and OnSessionEnd deal specifically with the starting and 

ending of sessions. OnError is another special function that provides code that should run if 

the application has an error. This is an ideal way to present "friendly" error messages to your 

users, while alerting you with pertinent details when there is a problem, such as sending out 

automatic warning emails or logging errors to a database. 

 



SDG ColdFusion Training, Session 4 

Page 17 of 21 

Within the Application.cfc, you can set several application properties, including the application 

name. These properties may be used to control specific behaviors within the application and 

can be reference anywhere within the Application.cfc. These are usually set at the start of the 

Application.cfc for code cleanliness and to make them available to all following functions. 

<cfcomponent> 

<cfset THIS.name="artGallery"> 

... 

</cfcomponent> 

Notice that these variables are scoped using the "this" keyword. "This" is commonly used in 

object-oriented programming to refer to "self', so here, this is referring to the application itself. 

Once an application variable is created, it will persist for a specified period. By default, this is 

two days of application inactivity, but you can override this default by setting a new timeout 

within the Application.cfc file using the ApplicationTimeout property and the CreateTimeSpan() 

function: 

<cfset THIS.applicationTimeout = CreateTimeSpan(0,0,0,1)>  

Here, we're setting the application timeout to 1 second. For production purposes, this is 

obviously impractical, but during development, this is useful in avoiding having to restart the 

server if a problem occurs while creating your initial application variables. The 

CreateTimeSpan() attributes are as follows: (days, hours, minutes, seconds). 

Because application variables are persistent across all page requests, they are ideal for storing 

frequently used, but infrequently changed, values. To create an application variable, you use 

the <cfset> tag with the Application prefix. One extremely common use of application variables 

is setting an application-wide datasource name, eliminating the need to maintain the value 

throughout your code and increasing code portability: 

<cfset application.dsn="CF8introArtGallery"> 

Note, that the Application prefix is always required. In general, you want to create your 

application variables when the application first initializes, so they are placed within the 

onApplicationStart event handler. 

<cfcomponent> 

<cffunction name="onApplicationStart"> 

<cfset application.dsn="CF8introArtGallery"> 

</cffunction> 

</cfcomponent> 

We will go more in-depth with using these various Application.cfc event handlers in the next 

session when we begin really looking into components. 

Session Framework 

First, a reminder on the importance of scoping: if a variable is not scoped, ColdFusion will 

attempt to figure out what you mean by searching for a variable with that name in specific 

scopes. In order, it will search these scopes (in order): 

1. Query result variables (where available) 

2. Local variables 

3. CGI variables 

4. File variables 

5. URL variables 

6. Form variables 

ColdFusion will never search the Session, Application, Server, or Client scopes unless they 

are specifically declared. So always scope your variables! 

Session variables store information specific to a single browser's interaction with a ColdFusion 

application, commonly called a “session”. Like application variables, session variables are 

stored in the server's memory, so they are lost if the server is restarted. They also can expire 

after a specified length of inactivity has passed. Session variables are generally used to hold 

information specific to each user, such as their login status, preferences, shopping cart, and so 

on. 

To enable your application to use session variables, you must set the application's 

SessionManagement and SetClientCookies properties to yes in the Application.cfc. You can 

also change the timeout value using the SessionTimeout attribute and the CreateTimeSpan() 

function. By default, session variables will time out in 20 minutes. If you wish to have them last 

longer, you must first raise the maximum time out value in your server's ColdFusion 

administrator. Each page request re-starts the session timeout. So, using the default setting, 

there can be up to 20 minutes of browser inactivity before the session variables become 

unavailable and new session variables must be created for the browser. 

In this example, we have enabled session variables in our application and set them to time out 

in 1 hour. 

<cfcomponent> 

<cfset THIS.name="artGallery"> 

<cfset THIS.sessionmanagement="yes"> 

<cfset THIS.setclientcookies="yes"> 

<cfset THIS.sessiontimeout="#CreateTimeSpan(0,1,0,0)#"> 

</cfcomponent> 

Session variables are unique to a particular web browser. By default, this is done through 

browser cookies. The first time a browser requests a page from a ColdFusion web application 

using session management, the server will generate a unique identifier for the browser 

comprised of two values: CFID and CFTOKEN. Two persistent cookies containing these 

values will be returned to the browser if they do not already exist, then for each subsequent 

page request, the cookies are sent to the server and verified against the values stored on the 

server. All other session variables are stored in the server memory using these two variables.  

This process ensures that variables created during a browser session are available to that 

browser during its session. By default, CFID and CFTOKEN are set as persistent cookies, so 

session variables remain in memory until the timeout is exceeded, even if the browser is 

closed and reopened. To change this so that the session ends if the browser is closed, you 

must select J2EE Sessions in the ColdFusion Administrator. 

Session variables are set the same as other variables, making sure to include the Session 

prefix: 

<cfset session.bgcolor ="red"> 

When session management is enabled, CFID and CFTOKEN are globally available as 

variables. If a browser has cookies disabled, your application will need to pass the values from 

page to page to maintain the session. You can do this manually, by appending the variables as 

dynamic URL parameters on every HTML hyperlink, as hidden values in every form, and by 

adding ADDTOKEN="yes" to all <cflocation> tags.  

Or, save yourself some headache, and wrap all URLs in your code with the 

URLSessionFormat() function, which will check to see if cookies are enabled. If not, it will 

rewrite URLs to include these variables as needed. 

<cfoutput> 

<a href="#URLSessionFormat(‘h.cfm’)#">Home</a> 

</cfoutput> 

All ColdFusion scopes are basically structures, a data type which hold multiple values as 

name=value pairs tied to a single variable name. To remove all of the values from a structure, 

you can use the StructClear() function, so to clear all session variables from the browser, you 

can use 

<cfset StructClear(session)> 

However, for best practices, you should never call this function on the session structure, as it 

will clear ALL session variables, including temporarily clearing the SessionID, CFID and 

CFTOKEN variables (though these are restored on the next call to the application). For 

session structures, you should instead selectively delete each individual variable using the 

StructDelete() function: 

<cfset StructDelete(session, "variablename")> 

Both of these functions can be used on other structures, including Form, Cookie, URL, and 

Application. However, StructClear cannot be used on the CGI scope or on queries. 

StructDelete can be used on queries, but the results may not be quite what you'd expect... 

Best Practices Reminders 
1.  When developing applications, also look for practical opportunities to reuse code to 

simplify coding and make maintenance easier. 

2. When using custom tags, remember attributes are stored in the Attributes scope; avoid 

relying on system specific variables (URL, Form, Session, etc) and on specific variables 

existing in the caller page. Instead, use attributes for maximum flexibility. 

3. If your custom tag passes back a value, use an attribute to allow the calling page to set 

the name of the returning variable, thereby avoiding unintentional variable overwriting. 

4. Always document custom tags, particularly denoting required/optional attributes 

5. A custom tag's first action should always be to check for required/optional attributes and 

respond appropriately when they are missing or malformed. 

6. Only declare an ending tag on a custom tag, or self-close it, if you specifically desire for 

ColdFusion to call the tag twice, one for start, once for end.  

7. Organizing custom tags by category/purpose can aid in reuse, and allows for the use of 

the <cfimport> function for cleaner code. 

8. When using ColdFusion components, arguments passed and values returned must 

match the types declared when creating the function. 

9. Use the Application.cfc framework for all new applications, and convert legacy apps 

from Application.cfm as soon as possible 

10. Never place passwords on the end of a URL, never save them as cookies, and never 

hard code them into the action page 

11. Never use StructClear on Session variables 

12. Wrap URLs in URLSessionFormat() to enable session information to be kept regardless 

of cookie existence 

